/*
Emily Thorson	
Proposal title: EFFECTS OF MISINFORMATION NEWS COVERAGE ON MEDIA TRUST


HYPOTHESES

Stated-Hyp1: "Increasing the salience of misinformation will decrease trust in news from social media.
(Respondents who answered either version of Q1 first will exhibit lower trust in social
media than those who answered Q2 first.)"

	Test-Hyp1: Trust in social media news will be lower among those who got misinformation treatment first vs. those who got it later.

Stated-Hyp2: "Increasing the salience of misinformation will increase trust in mainstream news
sources. (Respondents who answered either version of Q1 first will exhibit higher trust
in news from mainstream news sources than those who answered Q2 first.)"

	Test-Hyp2: Trust in mainstream news will be higher among those who got misinformation treatment first vs. those who got it later.

Stated-Hyp3: "Increasing the salience of misinformation on social media will increase trust in mainstream news sources. (Respondents who received the “social media” version of Q1 will
exhibit more trust in mainstream news sources than respondents who received the non-social
media version)"

	Test-Hyp3: The effect of social media treatment will be moderated by misinformation treatment. Trust in mainstream news will be higher among those who got social media treatment (vs. didn't get the social media treatment) and also got the misinformation treatment (vs. didn't get misinformation treatment)

Stated-Hyp4: "Increasing the salience of misinformation on social media will decrease trust in social
media news sources. (Respondents who received the “social media” version of Q1 will exhibit
less trust in social media news sources than respondents who received the non-social
media version.)"

	Test-Hyp4: The effect of social media treatment will be moderated by misinformation treatment. Trust in social media news sources will be lower among those who got social media treatment (vs. didn't get the social media treatment) and also got the misinformation treatment (vs. didn't get misinformation treatment)
 
	
********************************************************************************
NOTES

*/

clear all
use "ThorsonS42.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATION


* social media mentioned or not
	tab P_THORSON_A_B

	/*
	1.00	Show A Versions of Q3 and Q4 // social media mentioned
	2.00	Show B Versions of Q3 and Q4 - soc media not mentioned
	*/
	
	recode P_THORSON_A_B (1=1) (2=0), gen(socialmedia_mentioned)
	tab socialmedia_mentioned
	
* trust question first or second
	
	tab P_THORSON_Q5_Q6
	/*
	1.00	Q5 and Q6 shown second // trust questions second; misinformation salience first
	2.00	Q6 and Q5 shown first // 
	*/
	
	recode P_THORSON_Q5_Q6 (2=0) (1=1), gen(misinformation_salient)
	tab misinformation_salient
	
* OUTCOME MEASURES
	// dependent vars mentioned on pages 2 and 3 of proposal.

* trust in social media (facebook + twitter)

	gen facebook= Q6_2_THORSONC
	replace facebook= Q6_1_THORSONC if facebook==.
	
	gen twitter= Q6_2_THORSOND
	replace twitter=Q6_1_THORSOND if twitter==.
	replace facebook=. if facebook>5
	replace twitter=. if twitter>5
	gen trust_socialmedia= facebook+twitter
	tab trust_socialmedia
	
* trust in news (local+national newspapers+tv network)

 	gen local = Q6_2_THORSONA 
	replace local=Q6_1_THORSONA if local==.
	
	gen national=Q6_2_THORSONB
	replace national=Q6_1_THORSONB if national==.
	
	gen network=Q6_2_THORSONE
	replace network=Q6_1_THORSONE if network==.
	
	replace local=. if local>5
	replace national=. if national>5
	replace network=. if network>5
	gen trust_mainstreamnews= local+national+network
	tab trust_mainstreamnews

	
********************************************************************************

* ANALYSIS
	

*Test-Hyp1: Trust in social media news will be lower among those who got misinformation treatment first vs. those who got it later.

	reg trust_socialmedia i.misinformation_salient
	// do not reject. 0.001
	tess 1.misinformation_salient -, init(ThorsonS42) bonf(2)
	
* Test-Hyp2: Trust in mainstream news will be higher among those who got misinformation treatment first vs. those who got it later.

	reg trust_mainstreamnews i.misinformation_salient
	// reject. 0.120
	tess 1.misinformation_salient +, bonf(2)
	
*Test-Hyp3: The effect of social media treatment will be moderated by misinformation treatment. Trust in mainstream news will be higher among those who got social media treatment (vs. didn't get the social media treatment) and also got the misinformation treatment (vs. didn't get misinformation treatment)

	reg trust_mainstreamnews i.socialmedia_mentioned##i.misinformation_salient
	// reject. 0.166
	tess 1.socialmedia_mentioned#1.misinformation_salient +, bonf(2)
	
*Test-Hyp4: The effect of social media treatment will be moderated by misinformation treatment. Trust in social media news sources will be lower among those who got social media treatment (vs. didn't get the social media treatment) and also got the misinformation treatment (vs. didn't get misinformation treatment)
 
	reg trust_socialmedia i.socialmedia_mentioned##i.misinformation_salient
	// reject. 0.622	
	tess 1.socialmedia_mentioned#1.misinformation_salient -, bonf(2)
